

# Appendice 3

## Programmi per il calcolatore

1. Calcolo dei valori di una funzione
2. Calcolo di limiti finiti
3. Calcolo della derivata in un punto
4. Calcolo dell'integrale definito
5. Calcolo di un integrale con il metodo Montecarlo

In questa appendice presentiamo alcuni programmi per il calcolatore relativi agli argomenti svolti nel testo. Tutti i programmi sono scritti in linguaggio BASIC, in particolare nel BASIC dei personal computer che usano il sistema operativo MS DOS. Tuttavia, le istruzioni usate sono solo le più comuni del linguaggio, e perciò i programmi che presentiamo dovrebbero funzionare, senza modifiche, anche su altri calcolatori.

In questa appendice supponiamo che il lettore posseda già le nozioni elementari della programmazione in BASIC; chi non le avesse, può iniziare leggendo uno dei molti manuali di BASIC esistenti<sup>1</sup>.

## 1. Calcolo dei valori di una funzione

Nel linguaggio BASIC, come in molti altri linguaggi di programmazione, le funzioni elementari dell'analisi matematica sono già definite: potenze, esponenziale, logaritmo, funzioni trigonometriche, ecc. È quindi facile calcolare i valori di queste funzioni.

Ecco un esempio: calcolare i valori della funzione

$$y = \sin x$$

per  $x$  che varia tra 0 e 2, con incrementi di 0,1.

La variabile  $x$  assume dunque i valori  $x=0$ ,  $x=0,1$ ,  $x=0,2$ , ... fino a  $x=2$ ; il programma che esegue questo calcolo è:

```
10 FOR X=0 TO 2 STEP 0,1
20 PRINT X, SIN(X)
30 NEXT X
```

Avviando il programma con il comando RUN, appaiono sul video le coppie di valori  $x$ ,  $\sin x$ .

Abbiamo visto nel cap. 1 che dalle funzioni elementari si ottengono altre funzioni, ad esempio dalle potenze di  $x$  si ottengono i polinomi. Ecco come modificare il programma precedente per calcolare i valori del polinomio

$$y = 4x^3 + 5x - 2$$

corrispondenti agli stessi valori di  $x$  dell'esempio precedente:

```
10 FOR X=0 TO 2 STEP 0,1
20 PRINT X, 4*X^3+5*X-2
30 NEXT X
```

Si capisce da questi esempi che lo stesso programma permette di calcolare i valori di qualunque funzione: basta modificare l'istruzione 20, scrivendo nella PRINT la funzione che interessa.

In questo modo, però, si deve scrivere un programma per ogni funzione. È possibile, invece, scrivere un unico programma valido per più funzioni? Vediamo un esempio, scrivendo un programma adatto a tutti i polinomi. Iniziamo spezzando il programma in una serie di operazioni successive:

1. il programma chiede all'utente il grado del polinomio;
2. il programma chiede all'utente i coefficienti del polinomio;
3. il programma chiede all'utente l'intervallo in cui deve variare la  $x$ ;
4. il programma chiede all'utente l'incremento da dare alla  $x$ ;
5. il programma esegue i calcoli e mostra i risultati.

<sup>1</sup> Un'introduzione alla programmazione in BASIC è contenuta in E. Castelnuovo, C. Gori Giorgi, D. Valenti, *La matematica nella realtà*, vol. II e vol. III.



La fase 1 non necessita di ulteriori commenti; la fase 2, invece, si può spezzare ulteriormente:

- 2.1. il programma riserva in memoria lo spazio per i coefficienti;
- 2.2. il programma chiede all'utente i coefficienti e li memorizza.

Anche la fase 5 può essere ulteriormente decomposta:

- 5.1. ripetere per tutti i valori di  $x$  assegnati;
- 5.2. eseguire il calcolo del polinomio;
- 5.3. mostrare sul video il valore trovato.

Per eseguire i calcoli della fase 5.2, conviene osservare che, ad esempio, il polinomio

$$ax^2+bx+c$$

si può ottenere così:

- 5.2.1 si parte dal coefficiente del termine di grado massimo:  $a$ ;
- 5.2.2. si moltiplica per  $x$  e si aggiunge il coefficiente del grado successivo:  
 $(a)x+b=ax+b$ ;
- 5.2.3. si ripete l'operazione 5.2.2:  $(ax+b)x+c=ax^2+bx+c$ .

Se il polinomio è di grado maggiore di 2, ad esempio 5, basta ripetere 5 volte l'operazione 5.2.2.

Ora che il problema è stato scomposto in passi semplici, è facile scrivere il programma:

```

10  OPTION BASE 0
20  INPUT "GRADO DEL POLINOMIO": GRADO
30  DIM COEFF(GRADO)
40  FOR I=0 TO GRADO
50  PRINT "COEFFICIENTE";I;
60  INPUT COEFF(I)
70  NEXT I
80  INPUT "ESTREMO INFERIORE";INFER
90  INPUT "ESTREMO SUPERIORE";SUPER
100 INPUT "PASSO";PASSO
110 FOR X=INFER TO SUPER STEP PASSO
120 GOSUB 200
130 PRINT X,VAL
140 NEXT X
150 END
200 REM ***CALCOLO DEL POLINOMIO***
210 VAL=0
220 FOR I=GRADO TO 0 STEP -1
230 VAL=VAL*X+COEFF(I)
240 NEXT I
250 RETURN

```

L'istruzione 10 dice al calcolatore che, nella successiva istruzione DIM, la numerazione dei coefficienti inizia da 0 e non da 1; in questo modo, eseguendo ad esempio DIM COEFF(2) si ha spazio in memoria per 3 coefficienti:  $a_0$ ,  $a_1$ ,  $a_2$ .

Le istruzioni da 20 a 100 realizzano le fasi da 1 a 4. L'istruzione 110 corrisponde alla fase 5; la fase 5.2, e cioè il calcolo del polinomio, viene implementata dalla subroutine 200. Infine, l'istruzione 130 mostra i risultati.

## 2. Calcolo di limiti finiti

Iniziamo con l'esempio seguente: calcolare

$$\lim_{x \rightarrow +\infty} (1+0,5^x).$$

Possiamo valutare la funzione  $(1+0,5^x)$  per  $x=1$ ,  $x=2$ ,  $x=\dots$  ed osservare se i valori ottenuti tendono ad un limite finito.

Ecco un semplice programma<sup>1</sup> che esegue questo calcolo:

```
100  X=X+1
110  PRINT X,1+0,5^X
120  GO TO 100
```

Lanciando il programma, si vedono scorrere sul video i valori successivi della funzione, che si fanno sempre più prossimi a 1 e, dopo qualche secondo, arrivano esattamente a 1. Questo avviene quando  $x$  vale circa 30. Come si spiega questo risultato? Il numero

$$1+0,5^{30}$$

è certamente maggiore di 1, perché  $0,5^{30}$  è maggiore di zero. È però vero che

$$\lim_{x \rightarrow +\infty} (1+0,5^x)=1$$

ma, come sappiamo, la funzione  $1+0,5^x$  non raggiunge il limite 1 per nessun valore finito di  $x$ . Perché invece il calcolatore ha trovato esattamente 1? La spiegazione del risultato ottenuto con il programma è questa: il calcolatore lavora con numeri decimali limitati, in genere con 8 cifre dopo la virgola; quando il numero differisce da 1 solo per la nona cifra decimale, ad esempio 1,000000003, la macchina non "sente" la differenza e scrive 1.

Dunque la macchina permette di calcolare i limiti, ma solo in modo approssimato. In questa matematica approssimata, i limiti finiti possono essere confusi con i valori assunti dalla funzione: solo la mente umana riesce ad essere talmente sottile da distinguere tra i due concetti. Modifichiamo ora il programma per calcolare

$$\lim_{x \rightarrow -\infty} (1+0,5^x).$$

La modifica è immediata; si ha:

```
100  X=X-1
110  PRINT X,1+0,5^X
120  GO TO 100
```

Lanciando il programma, si vede che i valori della funzione crescono rapidamente, finché il programma si arresta e sul video compare il messaggio «OVERFLOW».

Questo risultato è dovuto al fatto che il calcolatore può lavorare solo con numeri compresi in un certo intervallo, ad esempio da  $10^{-39}$  a  $10^{39}$ . Quando un numero esce da questo intervallo, la macchina si ferma e segnala un «OVERFLOW», cioè un «trabocco».

Si capisce dunque che il calcolatore non permette di studiare i limiti infiniti.

<sup>1</sup> Questo programma non si arresta da solo: per fermarlo, occorre premere insieme i tasti CTRL e C.



### 3. Calcolo della derivata in un punto

La derivata di una funzione  $y=f(x)$  in  $a$  è il limite del rapporto incrementale

$$\frac{\Delta f}{\Delta x} = \frac{f(a+h)-f(a)}{h}$$

quando  $h$  tende a zero. Per scrivere un programma che calcoli la derivata di una funzione in un punto, si può calcolare il rapporto incrementale per  $h$  sempre più piccolo, e fermarsi quando i valori che si ottengono non variano più.

Si deve però fare attenzione alla divisione per zero: se  $h$  diventa talmente piccolo che la macchina non riesce a distinguerlo dallo zero, il rapporto incrementale ha uno zero al denominatore e non può essere più calcolato. I passi in cui si articola il programma sono

1. definire la funzione di cui si vuole calcolare la derivata in  $a$ ;
2. chiedere all'utente il valore di  $a$ ;
3. iniziare con un valore di  $h$  abbastanza grande, ad esempio  $h=1$ ;
4. calcolare il valore del rapporto incrementale;
5. ridurre  $h$ , ad esempio dimezzarlo;
6. verificare che  $h$  sia diverso da zero;
7. calcolare il nuovo valore del rapporto incrementale;
8. confrontarlo con il precedente:
  - 8.1. se la differenza è trascurabile: stampare il risultato e fermare il programma;
  - 8.2. se la differenza non è trascurabile: tornare al punto 5.

Ecco il programma che realizza questi passi; al punto 1 si è scelta la funzione  $y=xe^{-2x} \sin x$ :

```

100 TRASC=0.0001
110 DEF FNFUNZ(X)=X*EXP(-2*X)*SIN(X)
120 INPUT "VALORE DI A",A
130 H=1
140 RAPP=(FUNZ(A+H)-FUNZ(A))/H
150 H=H/2
160 IF H<TRASC THEN 300
170 NUOVORAPP=(FUNZ(A+H)-FUNZ(A))/H
180 IF ABS(RAPP-NUOVORAPP)<TRASC THEN 300
190 RAPP=NUOVORAPP
200 GO TO 150
300 PRINT "LA DERIVATA IN";A;"VALE";
    NUOVORAPP

```

L'istruzione 100 definisce che cosa si intende per "valore trascurabile": il numero TRASC viene usato nelle istruzioni 160 e 180 come se fosse il numero zero, e perciò viene detto **zero algoritmico**. Il significato è che quando un numero, come  $H$  o la differenza tra due rapporti incrementali, è minore dello zero algoritmico, allora è talmente piccolo da potersi trascurare.

Osserviamo che il programma calcola solo rapporti incrementali per  $h>0$ ; in altre parole, approssima la derivata con la differenza in avanti<sup>1</sup>. Se si vuole approssimarla con la differenza all'indietro, si deve porre  $H=-1$  nell'istruzione 130: tutti i valori successivi di  $H$ , ottenuti dividendo per 2, risulteranno negativi e dunque bisogna verificare anche le costruzioni 100 e 160.

<sup>1</sup> Vedi Complemento A al cap. 4.

Per valersi, invece, delle differenze centrali, basta modificare le istruzioni 140 e 170, sostituendo il rapporto incrementale con

$$\delta f = \frac{f(a+h) - f(a-h)}{2}.$$

Modificando poi l'istruzione 110, il programma permette di calcolare la derivata di altre funzioni.

#### 4. Calcolo dell'integrale definito

Nel Complemento A del cap. 6 sono stati esposti due metodi numerici per il calcolo approssimato di un integrale definito. Questi metodi sono molto adatti al calcolatore; vediamo quindi come programmarli, iniziando dal metodo dei rettangoli.

Si vuole dunque calcolare un integrale tra gli estremi  $a$  e  $b$ ; si divide allora l'intervallo  $[a, b]$  in  $n$  intervallini uguali di lunghezza

$$h = \frac{b-a}{n}$$

e in ogni intervallino si considera il punto medio:

$$x_1 = a + \frac{h}{2}, \quad x_2 = x_1 + h, \quad \dots, \quad x_n = x_{n-1} + h.$$

Infine si calcola la somma:

$$\Sigma_n = h[f(x_1) + f(x_2) + \dots + f(x_n)]$$

che approssima il valore dell'integrale tanto meglio, quanto più  $n$  è grande. Quindi i passi in cui si articola il programma sono:

1. definire la funzione di cui si vuole calcolare l'integrale;
2. chiedere all'utente i valori di  $a$ , di  $b$  e di  $n$ ;
3. calcolare  $h$ ;
4. calcolare la somma;
5. stampare il risultato.

Ecco il programma, scritto per la funzione  $y = x^2 \cos x$ :

```

100 DEF FNFUNZ(X)=X*X*COS(X)
110 INPUT "ESTREMO INFERIORE";A
120 INPUT "ESTREMO SUPERIORE";B
130 INPUT "NUMERO DI RETTANGOLI";N
140 H=(B-A)/N
150 FOR I=1 TO N
160 ARGOM=A+(2*I-1)*H/2
170 SOMMA=SOMMA+FUNZ(ARGOM)
180 NEXT I
190 PRINT "L'INTEGRALE VALE";H*SOMMA

```

Venendo ora al metodo dei trapezi, si devono considerare i punti:

$$x_1 = a + h, \quad x_2 = a + 2h, \quad \dots, \quad x_{n-1} = a + (n-1)h, \quad x_n = a + nh = b$$

e la formula:

$$T_n = h \left[ \frac{f(a)}{2} + f(x_1) + \dots + f(x_{n-1}) + \frac{f(b)}{2} \right].$$

Convien partire dallo stesso programma del metodo dei rettangoli, modificandolo per il metodo dei trapezi. Si deve:

- modificare il ciclo nell'istruzione 150, fermandolo ad  $I=N-1$ ;
- modificare la 160 in base alle nuove espressioni dei punti  $x_1, \dots, x_n$ ;
- aggiungere il primo e l'ultimo termine della somma  $T_n$ .



Ecco come si presenta il programma così modificato:

```

100 DEF FNFUNZ(X)=X*X*COS(X)
110 INPUT "ESTREMO INFERIORE";A
120 INPUT "ESTREMO SUPERIORE";B
130 INPUT "NUMERO DI INTERVALLI";N
140 H=(B-A)/N
150 FOR I=1 TO N-1
160 ARGOM=A+I*H
170 SOMMA=SOMMA+FUNZ(ARGOM)
180 NEXT I
190 SOMMA=SOMMA+FUNZ(A)/2+FUNZ(B)/2
200 PRINT "L'INTEGRALE VALE";H*SOMMA

```

Si consiglia di usare questi programmi per confrontare i due metodi di integrazione numerica; in genere, all'aumentare di  $n$  i due metodi danno risultati sempre più vicini.

## 5. Calcolo di un integrale con il metodo Montecarlo

La fig. 1 mostra un rettangolo che ha per base l'intervallo  $[a, b]$  e per altezza  $h$ ; il rettangolo è tagliato a metà dalla funzione costante

$$y = \frac{h}{2}.$$

Immaginiamo ora di gettare a caso su questo rettangolo una manciata di 100 "punti", ad esempio 100 chicchi di riso: quanti di questi cadranno nella parte colorata della figura, cioè sotto la funzione  $y = \frac{h}{2}$ ?

È chiaro che il fenomeno è casuale e quindi va studiato dal punto di vista della probabilità: visto che la parte colorata è la metà dell'intero rettangolo, ogni punto ha probabilità 0,5 di cadere nell'area colorata. Quindi è probabile che 50 punti su 100 cadano nell'area colorata.




Fig. 1




Fig. 2

La fig. 2 ripropone lo stesso problema, ma per la funzione

$$y = \frac{h}{4},$$

che individua il rettangolo colorato la cui area è  $\frac{1}{4}$  dell'area totale.

Ragionando come prima, ogni punto ha probabilità  $\frac{1}{4}$  di cadere nell'area colorata, e quindi è probabile che 25 punti su 100 cadano nell'area colorata.

Da questi due esempi si intuisce la regola generale: ogni punto ha una probabilità di cadere nell'area colorata data dal rapporto tra quest'area e l'area totale.

Immaginiamo ora di voler calcolare l'integrale definito tra gli estremi  $a$  e  $b$  della funzione

$$y=f(x)$$

disegnata in fig. 3; sappiamo che questo integrale misura l'area sotto la curva, che in figura è colorata. Scegliamo un numero  $h$  più grande del massimo assoluto della funzione nell'intervallo  $[a, b]$ , e consideriamo il rettangolo che ha per base tale intervallo e per altezza  $h$ .

Distribuiamo a caso 100 punti in questo rettangolo, e contiamo quanti ne cadono nell'area colorata (fig. 4). Se ve ne cadono, ad esempio, 32, possiamo dire che, probabilmente, l'area colorata è  $\frac{32}{100}$  dell'area del rettangolo. Cioè:

$$\int_a^b f(x)dx = 0,32h(b-a)$$




Fig. 3




Fig. 4

Dunque, questo ragionamento porta a scoprire quanto vale “probabilmente” un integrale definito; è chiaro che la valutazione è tanto più attendibile quanto maggiore è il numero di punti.

Questo modo di procedere è detto **metodo Montecarlo**, dal nome del famoso Casinò in cui, appunto, tutto è affidato alle leggi del caso.

Il metodo Montecarlo si presta bene ad essere programmato: basta ripetere molte volte l'operazione di scelta casuale di un punto, e poi andare a vedere se questo punto si trova sopra o sotto la curva.

Ecco il programma:

```

100 INPUT "ESTREMI DELL'INTERVALLO";A,B
110 INPUT "ALTEZZA DEL RETTANGOLO";H
120 INPUT "QUANTI PUNTI?";PUNTI
130 DEF FNFUNZ(X)=X*X
140 FOR I=1 TO PUNTI
150 X=A+RND(1)*(B-A)
160 Y=RND(1)*H
170 IF Y<=FUNZ(X) THEN N=N+1
180 NEXT I
190 PRINT "L'INTEGRALE VALE" H*(B-A)*N/PUNTI

```

Il programma è scritto per la funzione

$$y=x^2,$$

ma basta modificare l'istruzione 130 per applicarlo ad altre funzioni. La



linea 150 pone  $x$  uguale ad un numero casuale compreso tra  $a$  e  $b$ , mentre la linea 160 pone  $y$  uguale ad un numero casuale compreso fra 0 ed  $h$ ; così il punto  $(x, y)$  è scelto casualmente nel rettangolo.

Questa scelta viene ripetuta un numero di volte uguale al valore della variabile PUNTI; ogni volta, l'istruzione 170 verifica se l'ordinata del punto è minore di quella del punto corrispondente sulla curva, e quindi se il punto si trova al di sotto della curva. In caso affermativo, viene incrementata la variabile  $N$ . Così, al termine del ciclo,  $N$  è uguale al numero di punti che cadono al di sotto della curva.

Per poter assegnare il valore di  $h$ , occorre prima trovare quanto vale il massimo assoluto di  $f(x)$  in  $[a, b]$  (vedi cap. 5). Conviene dare ad  $h$  proprio questo valore, o un valore di poco superiore; altrimenti il rettangolo è troppo alto rispetto alla curva, e la precisione del calcolo diminuisce.

## Esercizi

1. Generalizzando il programma del paragrafo 1, pag. 655, per il calcolo dei valori di un polinomio, scrivere un programma per il calcolo dei valori di una funzione razionale.

*(Usare il programma del par. 1 come subroutine; fare attenzione alla divisione per zero).*

2. Scrivere un programma che calcoli le radici successive di un numero  $n$ :  $\sqrt{n}$ ,  $\sqrt{\sqrt{n}}$ , ... Si ottiene una successione di valori: a quale limite tende questa successione?
3. Scrivere un programma per il calcolo di

$$\lim_{x \rightarrow 0} \frac{\sin x}{x}.$$

4. Scrivere un programma per il calcolo della costante  $e$  di Nepero, sfruttando l'espressione (vedi Complemento A del cap. 2):

$$e = \lim_{n \rightarrow \infty} \left(1 + \frac{1}{n}\right)^n.$$

5. Scrivere un programma che accetti in ingresso il grado e i coefficienti di un polinomio, e mostri il grado e i coefficienti della derivata di quel polinomio.
6. Usando come subroutine il programma dell'esercizio precedente, scrivere un programma che mostri i coefficienti della derivata di una funzione razionale.
7. Scrivere un programma che ricerchi gli asintoti orizzontali di una curva.
8. Facendo attenzione al trabocco, scrivere un programma che ricerchi gli asintoti verticali di una curva.
9. Scrivere un programma che ricerchi gli asintoti obliqui di una curva.
10. Sfruttando come subroutine il programma per il calcolo della derivata visto nel paragrafo 3, scrivere un programma che determini in quali punti di un intervallo una funzione è crescente e in quali è decrescente.  
*(Scegliere  $N$  punti equidistanti nell'intervallo, e calcolare la derivata in quei punti).*
11. Scrivere un programma che ricerchi i punti di massimo, di minimo e di flesso di una curva.
12. Usando come subroutine i programmi degli esercizi 7-11, scrivere un programma che studi il grafico di una funzione.
13. Confrontare i risultati ottenuti con i programmi per il calcolo di un integrale definito presentati nei paragrafi 4 e 5.  
*(Applicare i tre programmi ad una funzione il cui integrale è noto, provando per valori diversi di  $N$ ).*
14. Scrivere un programma per il calcolo esatto dell'integrale indefinito di un polinomio.

